

Programme de calcul du tableau T1 (langage SAS)

```

*****
* Tableau 136 - Effectifs et prévalence d'individus ne pouvant pas sans aide réaliser      *
* les activités du chapitre 1 "Tha" selon milieu social                                *
***** ;

options fmtsearch=(hid99) nodate nonumber ;

data t1 ; merge hid99.modb_c (keep=ident numind btoil bhab1 bali1 bali2 bali3 in=x)
              hid99.modmen_c (keep=ident milieu1) ;
              by ident ;

              if x ;
              if milieu1 in ( ' ', '9') then milieu1='9' ;

data t2 ; merge t1 hid99.mindiv_c (keep=ident numind poidsco) ;
              by ident numind ;

              if bali3=' ' and bali1='0' then bali3='a' ;

              if btoil in ('0','5','6') then aide1=100 ; else aide1=0 ;
              if bhab1 in ('0','4','5','6') then aide2=100 ; else aide2=0 ;
              if bali1 in ('0','4','5') then aide3=100 ; else aide3=0 ;
              if bali2 in ('0','4') then aide4=100 ; else aide4=0 ;
              if bali3 in ('a','4') then aide5=100 ; else aide5=0 ;

proc summary data=t2 ;
class milieu1 ;
var poidsco ;
output out=fini sum=eff ;

proc summary data=t2 ;
class milieu1 ;
var aide1 aide2 aide3 aide4 aide5 ;
weight poidsco ;
output out=finp mean= ;

data fin; set fini finp ;
if milieu1=' ' then milieu1='E' ;

proc tabulate data=fin noseps ;
title "Tableau 136" ;
title2 "Population et prévalence d'individus ne pouvant pas sans aide réaliser" ;
title3 "les activités quotidiennes du chapitre 1 : toilette - habillage - alimentation" ;
title4 "selon le milieu social de la personne de référence du ménage" ;
footnote "c:\user\hid\progsas\incapacite\tab32\t32tha_mil.sas" ;
class milieu1 ;
var eff aide1 aide2 aide3 aide4 aide5;
table milieu1=' ', eff*F=10.0 (aide1 aide2 aide3 aide4 aide5)*F=9.2
/box="Milieu social de la personne de référence du ménage" condense misstext='/' rts=38 ;

keylabel all='Ensemble' sum=' ' ;
label eff="Population considérée"
      aide1="Faire habituellement leur toilette"
      aide2="S'habiller et se déshabiller"
      aide3="Couper leur nourriture"
      aide4="Se servir à boire"
      aide5="Boire et manger la nourriture prête" ;

format milieu1 $FMIL. ;
run ;

```

